U.S. DEPARTMENT OF COMMERCE
National Telecommunications and Information Administration
INSTITUTE FOR TELECOMMUNICATION SCIENCES
325 Broadway
Boulder, Colorado 80303

January 30, 1985

MEMORANDUM TO: Users of the ITS Irregular Terrain Model

From: George A. Hufford

Subject: Modifications

We want to thank colleagues at GE in Lynchburg, VA, for pointing out that version 1.2.1 (dated April, 1979) of the ITM (the Longley-Rice model) will still produce bad results on some line-of-sight paths when the model is used in the point-to-point mode. These seem to turn up most often on long paths at high frequencies.

To avoid these bad results we are now suggesting another slight change in the model. We will refer to the newly altered model as version 1.2.2 and date it September, 1984.

In the original report (Longley and Rice, 1968) we would set β equal to the angle $2\pi\Delta r/\lambda$ of equation (3.2) and then replace (3.3) by

$$\hat{\beta} = 4.1917\times10^{-5} f h_{e1} h_{e2}/d \quad \text{radians.}$$

If $\hat{\beta} < 1.57$ then

$$\beta = \hat{\beta}.$$

Otherwise

$$\beta = 3.14 - (1.57)^2/\hat{\beta}.$$

Note that normally (3.2) is used only when d is large enough that $\hat{\beta}$ does not exceed $\pi/3$. Thus in most cases the first formula above will be used and results will not have changed. There seem, however, to be some abnormal cases where this quantity becomes large, and then the second formula will prevent things from getting out of hand.

While we have the chance we should also like to suggest two further changes to our computer implementation (Hufford et al., 1982) of the model. First, there have been complaints that the program will sometimes abort with an "exponent too large" or "divide by zero" error. This behavior is machine dependent and happens, again, on some line-of-sight paths in the point-to-point mode. It has to do with an evaluation of the exponential function and can be easily corrected. We have also examined all other cases where the exponential is evaluated, and we are convinced that underflow and overflow cannot occur there unless some input parameter has a really ridiculous value.

The second change is deeper and has to do with one of the peripheral subroutines that prepares input for the model. It appears yet again for the case of the line-of-sight path in the point-to-point mode. In brief, we propose changing the definition of the elevation angle of the fictitious horizon so that the resulting geometry is more self-consistent. It turns out that this is the most radical of the modifications and on some paths it will change results by many decibels. The examples we have seen have all been towards what seem to be more reasonable values.

The three changes are easily implemented in the two subroutines ALOS and QLRPFL. The two attachments show explicitly how the new versions should appear, what lines should be changed, and what lines should be added. We have been informally testing the resulting program for some time now and so far have found no additional bugs. That, of course, is no guarantee, and we would appreciate knowing of any contrary experiences.

References

Hufford, G. A., A. G. Longley, and W. A. Kissick (1982), A guide to the use of the ITS irregular terrain model in the area prediction mode, NTIA Report 82-100.

Longley, A. G., and P. L. Rice (1968), Prediction of tropospheric radio transmission loss over irregular terrain--a computer method 1968, ESSA Tech. Report ERL 79-ITS 67.

```
      SUBROUTINE QLRPFL(PFL,KLIMX,MDVARX)
         DIMENSION PFL(5)
C
C        SETS UP AND RUNS THE LONGLEY-RICE MODEL IN THE POINT-TO-POINT
C          MODE USING THE TERRAIN PROFILE IN PFL.
C            PFL(1)=ENP, PFL(2)=XI, PFL(3)=Z(0),...
C
      COMMON/PROP/KWX,AREF,MDP,DIST,HG(2),WN,DH,ENS,GME,ZGND,
     X   HE(2),DL(2),THE(2)
         COMPLEX ZGND
      COMMON/PROPV/LVAR,SGC,MDVAR,KLIM
C
      DIMENSION XL(2)
C
       DIST=PFL(1)*PFL(2)
       NP=PFL(1)
       CALL HZNS(PFL)
C         FIND DELTA H
       DO 11 J=1,2
   11  XL(J)=AMIN1(15.*HG(J),0.1*DL(J))
       XL(2)=DIST-XL(2)
       DH=DLTHX(PFL,XL(1),XL(2))
C         FIND EFFECTIVE HEIGHTS HE
       IF(DL(1)+DL(2) .LT. 1.5*DIST) GO TO 25
C         LINE-OF-SIGHT
          CALL ZLSQ1(PFL,XL(1),XL(2),ZA,ZB)
          HE(1)=HG(1)+DIM(PFL(3),ZA)
          HE(2)=HG(2)+DIM(PFL(NP+3),ZB)
          DO 21 J=1,2
   21     DL(J)=SQRT(2.*HE(J)/GME)*EXP(-0.07*SQRT(DH/AMAX1(HE(J),5.)))
          Q=DL(1)+DL(2)
          IF(Q .GT. DIST) GO TO 23
          Q=(DIST/Q)**2
          DO 22 J=1,2
          HE(J)=HE(J)*Q
   22     DL(J)=SQRT(2.*HE(J)/GME)*EXP(-0.07*SQRT(DH/AMAX1(HE(J),5.)))
   23     DO 24 J=1,2                                                   *REPLACE
          Q=SQRT(2.*HE(J)/GME)                                          *INSERT
   24     THE(J)=(0.65*DH*(Q/DL(J)-1.)-2.*HE(J))/Q                      *INSERT
       GO TO 28                                                         *INSERT
C         TRANSHORIZON
   25     CALL ZLSQ1(PFL,XL(1),0.9*DL(1),ZA,Q)
          CALL ZLSQ1(PFL,DIST-0.9*DL(2),XL(2),Q,ZB)
          HE(1)=HG(1)+DIM(PFL(3),ZA)
          HE(2)=HG(2)+DIM(PFL(NP+3),ZB)
   28  CONTINUE
C
       MDP=-1
       LVAR=MAX0(LVAR,3)
       IF(MDVARX .LT. 0) GO TO 31
          MDVAR=MDVARX
          LVAR=MAX0(LVAR,4)
   31  IF(KLIMX .LE. 0) GO TO 32
          KLIM=KLIMX
          LVAR=5
   32  CONTINUE
C
       CALL LRPROP(0.)
C
       RETURN
      END
```

```
      FUNCTION ALOS(D)
C          THE *LINE-OF-SIGHT ATTENUATION* AT DISTANCE D
C          A CONVEX COMBINATION OF PLANE EARTH FIELDS AND
C             DIFFRACTED FIELDS
C          A CALL WITH D=0 SETS UP INITIAL CONSTANTS
C
      COMMON/PROP/KWX,AREF,MDP,DIST,HG(2),WN,DH,ENS,GME,ZGND,
     X   HE(2),DL(2),THE(2)
         COMPLEX ZGND
      COMMON/PROPA/DLSA,DX,AEL,AK1,AK2,AED,EMD,AES,EMS,DLS(2),DLA,THA
C
      COMMON/SAVE/WLS,SAVE(49)
C
      COMPLEX R
C
      ABQ(R)=REAL(R)**2+AIMAG(R)**2
C
       IF(D .GT. 0.) GO TO 10
C
       WLS=0.021/(0.021+WN*DH/AMAX1(10E3,DLSA))
       ALOS=0.
       GO TO 80
C
   10  CONTINUE
       Q=(1.-0.8*EXP(-D/50E3))*DH
       S=0.78*Q*EXP(-(Q/16.)**0.25)
       Q=HE(1)+HE(2)
       SPS=Q/SQRT(D**2+Q**2)
       R=(SPS-ZGND)/(SPS+ZGND)*EXP(-AMIN1(10.,WN*S*SPS))                 *REPLACE
       Q=ABQ(R)
       IF(Q .LT. 0.25 .OR. Q .LT. SPS) R=R*SQRT(SPS/Q)
       ALOS=EMD*D+AED
       Q=WN*HE(1)*HE(2)*2./D
       IF(Q .GT. 1.57) Q=3.14-2.4649/Q                                   *INSERT
       ALOS=(-4.343*ALOG(ABQ(CMPLX(COS(Q),-SIN(Q))+R))-ALOS)*WLS+ALOS
C
   80  RETURN
      END
```